Maplet „Vyšetření kvadrik“

Pro podporu výuky křivek druhého stupně a jako doplněk této knihy naprogramoval student Pedagogické fakulty Jihočeské univerzity Marek Dvorožňák v roce 2008 tento maplet provádějící úplnou klasifikaci kvadrik krok za krokem.



Obrázek 3.4: Maplet *Vyšetření kvadrik*

Aplikace je dostupná zdrojovém CD knihy

Maplety mohou být spuštěny nezávisle na běhu programu Maple. Bohužel je však nutné, aby byl program Maple (verze 9.5 nebo vyšší) na počítači nainstalován. Maplety využívají jeho výpočetní jádro.

3.3 Modely vybraných ploch v Maple

Tato příloha je věnována užití Maple k odvození parametrických rovnic a grafickému znázornění vybraných kvadratických ploch. Uvedená řešení v Maple jsou inspirována příklady, které najdeme v knize na následujících stranách: str. 76 (rotační jednodílný hyperboloid), str. 90 (hyperbolický paraboloid) a str. 97 (rotační kuželová plocha).

1. Rotační jednodílný hyperboloid

Příklad: *Ukažte, že rotací přímky kolem osy, která je s přímkou mimoběžná, vznikne rotační jednodílný hyperboloid (Str. 76).*

Zobrazení v Maple:

```
> restart; with(linalg):
```

Úlohu budeme řešit nejprve obecně. Proto zadáme bod tvořící přímky `A` a její směrový vektor `u` jako obecné uspořádané trojice:

```
> A:=vector(3): u:=vector(3):
```

Tvořící přímku zadáme parametricky:

```
> p:=A+t*u;
```

```
> p:=evalm(p);
```

Potom definujeme matici rotace kolem osy z

```
> Rot_z:=matrix([[cos(s),-sin(s),0],[sin(s),cos(s),0],[0,0,1]]);
```

a vynásobíme s ní parametricky dané souřadnice přímky `p`

```
> ph:=evalm(Rot_z(s)&*p);
```

Výsledkem je parametrické vyjádření plochy, která vznikne příslušnou rotací přímky `p`.

$$ph = [(A_1+tu_1)\cos s-(A_2+tu_2)\sin s, (A_1+tu_1)\sin s+(A_2+tu_2)\cos s, A_3+tu_3]$$

Po dosazení konkrétních souřadnic bodu `A` a vektoru `u`

```
> A:=[3,0,0]; u:=[-1,2,5];
```

můžeme přistoupit k zobrazení tvořící přímky:

```
> plot3d(p,t=-2..1,k=-1..1,axes=normal,labels=[x,y,z],
scaling=constrained,view=[-4.2..4.2,-4.2..4.2,-4..6],
tickmarks=[3,3,3],thickness=3);
```

případně, s pomocí příkazu `seq`, celé posloupnosti jejích poloh během rotace kolem osy z:

```
> ph:=map(unapply,ph,s,t):
```

```
> plot3d([seq(evalm(ph(s/5,t)),s=0..30)],t=-2..1,k=-1..1,axes=normal,
labels=[x,y,z],scaling=constrained,view=[-4.2..4.2,-4.2..4.2,-4..6]);
```

Nakonec zobrazíme výslednou plochu - jednodílný rotační hyperboloid:

```
> plot3d(ph(s,t),s=0..2*Pi,t=-6..1,axes=frame,labels=[x,y,z],
scaling=constrained,view=[-4.2..4.2,-4.2..4.2,-4..6],tickmarks=[3,3,3],
orientation=[-40,70]);
```



Obrázek 3.5: Výsledný rotační jednodílný hyperboloid

2. Hyperbolický paraboloid

Příklad: *V daném prostorovém čtyřúhelníku $ABCD$, který leží na hyperbolického paraboloidu sestrojte přímky* 1. *a* 2. *regulu, obr. 2.28, 2.29 (Str. 90).*

Zobrazení v Maple:

Vyjdeme z obrázku 2.28 na straně 90. Nejprve definujeme přímky (úsečky) AD, BC:

```
> restart;
```

```
> p:=A+t*u; q:=B+t*v;
```

Každá přímka 1. regulu, určeného přímkami AB, CD, se dá vyjádřit parametricky následující rovnicí, kde `s` je reálný parametr:

```
> X:=p+s*(q-p);
```

Po dosazení konkrétních hodnot do proměnných `A`, `B`, `u` a `v`:

```
> A:=[0,0,0]: u:=[5,0,2]: B:=[0,25,10]: v:=[5,0,-2]:
```

definujeme obrazy úseček AD, BC (`usecky`) a plochy jimi určené (`plocha`):

```
> usecky:=plot3d(evalm(p),evalm(q),t=0..5,k=-1..1,thickness=6):
```

```
> plocha:=plot3d(evalm(X),t=0..5,s=0..1):
```

Nakonec úsečky i plochu vykreslíme příkazem `display`:

```
> plots[display](usecky,plocha,scaling=constrained);
```



Obrázek 3.6: Výsledný hyperbolický paraboloid

3. Rotační kuželová plocha

Příklad: *Napište rovnici kuželové plochy s vrcholem v bodě* $V = [3, 0, 0]$, *jejíž tvořící přímky svírají s osou* x *úhel* $\varphi = 30°$ *(Str. 97).*

Zobrazení v Maple:

Jednotlivé kroky řešení se objevily v řešeních předcházejících úloh. Proto je již nebudeme komentovat.

```
> restart;
> X:=V+t*u;
> V:=[3,0,0]; u:=[-3,0,sqrt(3)];
> plot3d(evalm(X),t=0..1,k=-1..1,axes=normal,thickness=5,
tickmarks=[3,3,3]);
```



Obrázek 3.7: Část tvořící přímky

```
> Rot:=matrix([[1,0,0],[0,cos(s),-sin(s)],[0,sin(s),cos(s)]]);
> Xh:=evalm(Rot&*X);
> plot3d(Xh,s=-Pi..Pi,t=0..1,scaling=constrained,axes=frame,
tickmarks=[3,3,3],color=grey,light=[90,-5,1,1,1]);
```



Obrázek 3.8: Část výsledné kuželové plochy

3.4 Řešení Příkladu 1 (str.55) v programu Maple

Zadání: Vyšetřete kvadriku [2], [5]

$$7x^2 + 6y^2 + 5z^2 - 4xy - 4yz - 22x + 24y + 2z + 30 = 0. \tag{3.1}$$

Kompletní kód řešení v Maple:

Poznámka: Výstupy některých příkazů kódu nejsou v textu uvedeny. Většinou se jedná o případy, kdy výstup pouze kopíruje zadaný příkaz.

```
> restart;
> with(LinearAlgebra): with(linalg): with(plots):
```

Obecnou rovnici kvadriky můžeme zapsat a vytvořit užitím matice kvadriky `K`:

```
> X:=Vector[row]([x,y,z,1]);
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} \mathrm{a}(1,\,1) & \mathrm{a}(1,\,2) & \mathrm{a}(1,\,3) & \mathrm{a}(1,\,4) \\ \mathrm{a}(1,\,2) & \mathrm{a}(2,\,2) & \mathrm{a}(2,\,3) & \mathrm{a}(2,\,4) \\ \mathrm{a}(1,\,3) & \mathrm{a}(2,\,3) & \mathrm{a}(3,\,3) & \mathrm{a}(3,\,4) \\ \mathrm{a}(1,\,4) & \mathrm{a}(2,\,4) & \mathrm{a}(3,\,4) & \mathrm{a}(4,\,4) \end{bmatrix}$$

```
> Kvadrika:=sort(expand(X.K.Transpose(X)),[x,y,z])=0;
```

$$\mathit{Kvadrika} := \mathrm{a}(1,\,1)\,x^2 + 2\,\mathrm{a}(1,\,2)\,x\,y + 2\,\mathrm{a}(1,\,3)\,x\,z + \mathrm{a}(2,\,2)\,y^2 + 2\,\mathrm{a}(2,\,3)\,y\,z$$
$$+\mathrm{a}(3,\,3)\,z^2 + 2\,\mathrm{a}(1,\,4)\,x + 2\,\mathrm{a}(2,\,4)\,y + 2\,\mathrm{a}(3,\,4)\,z + \mathrm{a}(4,\,4) = 0$$

Rovnice kvadriky dle zadání:

```
> RovKv:=7*x^2+6*y^2+5*z^2-4*x*y-4*y*z-22*x+24*y+2*z+30=0;
```